Designing a Gaussian Band Pass Filter

by Kevin G. Faison

The Filters discussed in Application Notes 4, 31 and 32 were of the Chebyshev or Butterworth response type. These responses are proven performers and are usually quite realizable. However, there are many other filter responses possible that may be more desirable for certain applications. The User Defined Response feature of FAISYN lets the designer realize a variety of specialized filters by permitting direct entry of NLP element values.

Gaussian response filters are often used in the IF's of certain receivers because of their favorable pulse response characteristics. In this example a Gaussian band pass filter will be designed to illustrate the user-defined response feature of FAISYN. The frequency response of the filter will be verified through a MMICAD simulation. In addition, a SPICE time domain simulation will be performed to investigate the response of the filter to a 50nS RF pulse. And finally, the pulse response of the Gaussian filter will be compared to that of a 0.1dB Chebyshev of equal 3dB bandwidth.

Design Procedure

All the filters synthesised by FAISYN are based on normalized low pass (NLP) prototype circuits. By allowing direct entry of NLP element values FAISYN enables the user to design just about any type filter for which a table of NLP element values exists. These would include the Bessel, Linear phase with equiripple, Gaussian to 3, 6 and 12dB, and the Lengendre response filters. In addition, there are the hybrid response filters which are not based on a particular polynomial but have instead been computer optimized to give certain performance characteristics. In any case, as long as a table of normalized lowpass element values exists for the response, FAISYN can be used to perform all the impedance and frequency transformations necessary to realize the final filter circuit.. This capability will now be demonstated through the design of a Gaussian band pass filter. The requirements for this design are summerized below:

Circuit Specification

Source Resistance = 50 Ohms
Load Resistance = 50 Ohms
Number of Poles N=3
Response = Gaussian
Lower 3dB frequency = 140 MHz
Upper 3dB frequency = 160 MHz

Filter Synthesis Using FAISYN

Figure 1 shows the table of NLP elements for the Gaussian response filter. It is important to remember that FAISYN assumes the elements are normalized to a radian cutoff frequency of 1 rad/Hz and a source resistance of 1 Ohm. This is usually not a problem since most of the tables in the literature follow this convension. However, for the case where the elements are normalized to the load, one simply enters the elements in reverse order (ie enter G(3) from table as G(1) etc.).

Gaussian Response

Rs	G(1)	G(2)	G(3)	RL
1	0.2624	0.8167	2.2262	1

Figure 1 Normalized low pass element values for N=3, Gaussian filter.

As in previous examples, the first step is to enter the source and load resistances and the filter order. The user will then choose the option to enter the Prototype Circuit from Table. FAISYN will prompt the user for each of the NLP element values, ie, G(1) through G(3). The equal source and load resistances allows us to choose an inductor as the leading element in the NLP circuit. The user may recall from an earlier example that this will result in a series LC as the first resonator in the bandpass circuit. The rest of the circuit specifications are entered exactly as in previous examples. The NLP prototype Gaussian to 6dB circuit and it's bandpass counterpart is shown in Figure 2 and 3.



Figure 2 The Gaussian NLP prototype filter

MMICAD Simulation

The MMICAD file written by FAISYN is given in Figure 4. The frequency response of the filter is shown in the MMICAD plot of Figure 5.

```
!=====FILTER SYNTHESIS PROGRAM V1.0 ======
!
!
! USER DEFINED Band Pass Filter
! N=  3
!  Lower 3dB frequency=  140.0000 MHZ
!  Upper 3dB frequency=  160.0000 MHZ
! FILENAME= GBPF.CKT
!
!======(c) 1992: OPTOTEK / K.FAISON======

GLOBAL
 DIM  FREQ=1E6 CAP=1E-12 IND=1E-9

VAR
 RSOURCE= 50
 RLOAD= 50

CKT
 SLC 1  2 L=104.4056 C=10.8310
 PRLC 2  0 L=8.6998 C=129.9818
 SLC 2  3 L=885.7769 C=1.2766
DEF2P 1 3  FILTER

TERM
 FILTER RSOURCE RLOAD

FREQ
 SWEEP  100  200 .5

OUT
 FILTER DB[S21] TRANS1
 FILTER DB[S11] TRANS1 R
 FILTER DB[S21] TRANS2
 FILTER DB[S11] TRANS2 R
 FILTER S11 SMITH
 FILTER SPAR SPAR

GRID
 RANGE   100  200   10
 TRANS1 -50 0 10 R -30 0
 RANGE   140.0000  160.0000    5.0000
 TRANS2 -5  0 0.5 R -20 0

LABEL
Gaussian Bandpass Filter
```

Figure 4 MMICAD simulation file



Figure 5 Frequency response of Gaussian band pass filter using MMICAD

SPICE Simulation

As mentioned previously the user has the option of creating MMICAD or SPICE simulation files when using FAISYN. A Spice simulation file for the current example is given in Figure 6. This file is set up for an AC analysis only. One should also note that the amplitude of the AC source is set to the ratio (Rs+RL)/RL so as to normalize the frequency response of the filter. Figure 7 shows the frequency response of the filter as determined by SPICE. It is comforting to see this agrees exactly with the MMICAD result.

```
*===== FILTER SYNTHESIS PROGRAM V1.0 =====
*
*
* USER DEFINED Band Pass Filter
* N=  3
*   Lower 3dB frequency=  140.0000  MHZ
*   Upper 3dB frequency=  160.0000  MHZ
* FILENAME= D:\PSPICE\GBPF.CIR
*
*  PLOT V( 3 ) FOR FILTER RESPONSE
*
*  NOTE: The AC magnitude of Vin is scaled to normalize
*     the filter response and is equal to (RL+Rs)/RL.
*
*======(c) 1992: OPTOTEK / K.FAISON=======

 Vin 100 0 AC        2.0000
 Rs 100 1  50

 L1 1  102   104.4056 nH
 C1 102  2    10.8310 pF
 L2 2  0     8.6998 nH
 C2 2  0   129.9818 pF
 L3 2  104   885.7769 nH
 C3 104  3     1.2766 pF

 RL  3  0  50

.AC LIN 51   109.6663 MEGHZ  189.6663 MEGHZ
.END
```

Figure 6 **SPICE** simulation file



Figure 7 Frequency response of Gaussian band pass filter using **SPICE**

Pulse Response

In order to investigate the respone of the filter to a 50nS (pulse width=1/filter 3dB bandwidth) RF pulse a SPICE transient analysis must be performed. The circuit file is modified to include the pulsed input source. To make things a little more interesting a third order, 0.1dB, Chebyshev filter is also added for comparison purposes. The file is given in Figure 8.

```
==== FILTER SYNTHESIS PROGRAM V1.0 ====
* This file has been modified to investigate the pulse
* response of the Gaussian to 6dB band pass filter.
* Two subcircuits have been added.  The first is a 3-pole
* 0.1dB Chebyshev filter that will be used for comparison
* purposes, the second sub-circuit is used to generate the
* pulsed RF input waveform.
*
* USER DEFINED Band Pass Filter
* N=  3
*   Lower 3dB frequency=  140.0000  MHZ
*   Upper 3dB frequency=  160.0000  MHZ
* FILENAME= D:\PSPICE\GBPF.CIR
*
*  PLOT V( 3 ) FOR FILTER RESPONSE
*
*  NOTE: The AC magnitude of Vin is scaled to normalize
*    the filter response and is equal to (RL+Rs)/RL.
* ========(c) 1992: OPTOTEK / K.FAISON =======

* Vin 100 0 AC        2.0000
 XRFpulse 100 RFpulse
 Rs 100 1  50

 L1 1  102   104.4056nH
 C1 102  2    10.8310pF
 L2 2  0     8.6998nH
 C2 2  0   129.9818pF
 L3 2  104   885.7769nH
 C3 104  3     1.2766pF

 RL  3  0  50

Rs2 100 4 50
XCBPF 4 5 CBPF
RL2 5 0 50
```

```
.SUBCKT CBPF 1 3
*===================================
* This sub-circuit contains a 3 pole Chebyshev filter
* The pulse response of this filter will be compared
* with the Gaussian example.
*
* Chebyshev Band Pass Filter
* N=  3 , ripple=0.1dB
*   Lower 3 dB frequency=  140.0000  MHZ
*   Upper 3 dB frequency=  160.0000  MHZ
* FILENAME= CBPF.CIR
*
*  PLOT V( 3 ) FOR FILTER RESPONSE
*
*  NOTE: The AC magnitude of Vin is scaled to normalize
*    the filter response and is equal to (RL+Rs)/RL.
*
*===================================

 L1 1  102   570.1055nH
 C1 102  2     1.9835pF
 L2 2  0     4.4582nH
 C2 2  0   253.6498pF
 L3 2  104   570.1055nH
 C3 104  3     1.9835pF
.ENDs

.SUBCKT RFpulse 1
*===================================
* This sub-circuit generates a pulsed RF waveform
* by multiplying a 50nS pulse with a 150MHz carrier.
*
*===================================
Vrf 100 0 sin(0 1 150MEGHz)
Vpulse 200 0 PULSE(0 2 0nS 1nS 1nS 50nS 1S)
R1 100 0 1G
R2 200 0 1G
Erfpulse 1 0 POLY(2) (100,0) (200,0) 0 0 0 0 1
Rs 1 0 1G
.ENDs

.TRAN 2nS 200nS 0 .35nS
*.AC LIN 51   100MEGHZ  200MEGHZ
.END
```

Figure 8 **SPICE** file modified for transient analysis

The **SPICE** plots of Figures 9 through 12 clearly demonstrate the superior time sidelobe suppression of the Gaussian response filter. This has important implications for pulse systems such as radar where large time sidelobes could mask hostile targets.



Figure 9 50nS RF input pulse



Figure 10 Pulse response of Gaussian filter



Figure 11 Pulse response of 0.1dB Chebyshev filter



Figure 12 Pulse responses of Gaussian and Chebyshev filters plotted in dB. The 1st time sidelobe of the Gaussian filter is approximately 40dB lower than that of the Chebyshev.